
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.6.2.11 which is part of the LTE_A test suite based on ‘iwd-TTCN3-B2013-03_D14wk37’ delivery

The test case can be demonstrated to run with one LTE UE (see section 5). Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_8_6_2_11
Test Group:
LTE_A_IWD_14wk37
ATS Version:
iwd-TTCN3-B2013-03_D14wk37
System Simulator used:
Anite Conformance Toolset solution
UE’s used:
Samsung Shannon UE

Verification Status:
PASS


4. General correction required for test case 8.6.2.11
Change 1
	Function Name
	f_TC_8_6_2_11_EUTRA

	Reason for change
	1. SIB5 configuration for cell 12 was missing in TTCN 
2. Step 15 is not implemented in TTCN which is listed in prose.
3. In function “f_EUTRA_SS_PdcpHandoverCtrl” target cell ID needs to be passed in the function.
4. As “v_TimerValue1” and “v_TimerValue2” are 30s plus tolerance and “p_LoggingInterval_r10” is ‘ms10240’ so UE can do multiple loggings of both the cells. Hence, “logMeasReport” measurements did on Cell 1 and Cell 12 UE can send multiple reports in “UEInformationReponse” in the list “logMeasInfoList-r10”

	Summary of change
	1. SIB5 configuration for cell 12 is added in TTCN. A new common function is created for its configuration, function “fl_InitSIB5Info_eMDT” 
2. “v_CellPowerList_AtT2” is added at step 15.
3. “eutra_Cell2” is passed in function “f_EUTRA_SS_PdcpHandoverCtrl” instead of cell 1.
4. Updated TTCN “UEInformationReponse” message to expect multiple values for “logMeasInfoList-r10”
Note: Please refer Prose CR R5-145549 for change 4

	Source of change
	RRC_MDT_Logged.ttcn


Before:
	  function f_TC_8_6_2_11_EUTRA() runs on EUTRA_PTC

  { //Logged MDT / Logging and reporting / Reporting at intra LTE handover / PLMN list

    var template (value) CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) CellPowerList_Type v_CellPowerList_AtT2;

    var Frequency_IE_Type v_Frequency_IE_f1;

    var Frequency_IE_Type v_Frequency_IE_f2;

    var EUTRA_CellInfo_Type v_CellInfo_Cell1;

    var EUTRA_CellInfo_Type v_CellInfo_Cell12;

    var CellIdentity v_CellIdentity_Cell1;

    var CellIdentity v_CellIdentity_Cell12;

    var PLMN_Identity v_PLMN_Identity_Cell1;

    var PLMN_Identity v_PLMN_Identity_Cell12;

    var PhysCellId v_PhysCellId_Cell1, v_PhysCellId_Cell12;

    var float v_TimerValue1;

    var float v_TimerValue2;

    var B48_Type v_AbsoluteTimeInfo;

    // var template (value) MobilityControlInfo v_MobilityControlInfo;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    v_CellPowerList_AtT0 := {

      cs_CellPower (eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell12, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_AtT1 := {

      cs_CellPower (eutra_Cell1, -79)

    };

    v_CellPowerList_AtT2 := {

      cs_CellPower (eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),

      cs_CellPower (eutra_Cell2, -79)

    };

    // Init variables

    f_EUTRA_Init(c3);

    v_Frequency_IE_f1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);

    v_Frequency_IE_f2 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell12);

    v_CellInfo_Cell1 := f_EUTRA_CellInfo_Get(eutra_Cell1);

    v_PhysCellId_Cell1 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);

    v_CellIdentity_Cell1 := v_CellInfo_Cell1.CellIds.CellIdentity;

    v_PLMN_Identity_Cell1 := v_CellInfo_Cell1.NAS_Parameters.Guti_Parameters.PLMN_Identity;

    v_CellInfo_Cell12 := f_EUTRA_CellInfo_Get(eutra_Cell12);

    v_CellIdentity_Cell12 := v_CellInfo_Cell12.CellIds.CellIdentity;

    v_PLMN_Identity_Cell12 := v_CellInfo_Cell12.NAS_Parameters.Guti_Parameters.PLMN_Identity;

    v_PhysCellId_Cell12 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell12);

    v_TimerValue1 := f_EUTRA_SetTimerToleranceMax(eutra_Cell12, nonProtocolTimer, 30.0);

    v_TimerValue2 := f_EUTRA_SetTimerToleranceMax(eutra_Cell1, nonProtocolTimer, 30.0);

    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -79);

    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell2, -79);

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    f_EUTRA_CellConfig_Def(eutra_Cell12);

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);

    //The UE is registered on PLMN1 (Cell 1) using the procedure described in TS 36.508[18]

    //clause 4.5.2.3 except that the ATTACH ACCEPT message indicates PLMN2 in

    //the Equivalent PLMN list as described in Table  8.6.2.11.3.3-5

    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell12);

    //Bring UE to inital state with Equivalent PLMN2

    f_EUTRA_IdleUpdated_WithEplmn (eutra_Cell12,

                                   PREAMBLE,

                                   cs_PLMN_List_1PLMN(f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell1))));

    f_EUTRA_RbEst_Def(eutra_Cell12);

    f_EUTRA_TestBody_Set (true);

    //@siclog "Step 1" siclog@

    //The SS transmits a LoggedMeasurementConfiguration message on Cell 12.

    v_AbsoluteTimeInfo := f_MDT_GetAbsolute_TimeStamp();

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell12,

                                 cs_TimingInfo_Now,

                                 cs_LoggedMeasurementConfiguration_with_v1080_IEs(v_PLMN_Identity_Cell12, '0000EF'O, '001A'O, '05'O, v_AbsoluteTimeInfo, -, -, -,

                                                                                  cs_LoggedMeasurementConfiguration_v1080_IEs({v_PLMN_Identity_Cell1, v_PLMN_Identity_Cell12}))));

    //@siclog "Step 2" siclog@

    //The SS transmits an RRCConnectionRelease message to release the RRC connection.

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell12);

    //@siclog "Step 3" siclog@

    //Wait 30 secods for UE performing the logging at regular time intervals on Cell 12.

    f_Delay(v_TimerValue1);

    //@siclog "Step 4" siclog@

    //The SS changes Cell 1 and Cell 12 levels according to the row "T1" in table 8.6.2.11.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 5" siclog@

    //The generic test procedure in TS 36.508 subclause 6.4.2.7 is performed on Cell 1.

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_EUTRA_TrackingAreaUpdate(eutra_Cell1);

    //@siclog "Step 6" siclog@

    //Wait 30 secods for UE performing the logging at regular time intervals on Cell 1

    f_Delay(v_TimerValue2);

    //@siclog "Step 7 - 14" siclog@

    // Steps 2 to 9 of the generic radio bearer establishment procedure in TS 36.508 subclause 4.5.3.3 are executed to successfully complete the service request procedure on Cell 1.

    f_EUTRA_RbEst_Def(eutra_Cell1);

    //Get security parameters

    v_Auth_Params := f_EUTRA_Security_Get ();

    //Inform the SS about the HO and about the source cell id

    f_EUTRA_SS_PdcpHandoverCtrl(eutra_Cell1, cas_PdcpHandoverInit_REQ(eutra_Cell1, eutra_Cell2));

    //Configure C-RNTI based contention resolution in cell 2

    f_EUTRA_SS_ConfigRachProcedure_HO (eutra_Cell2, cs_TimingInfo_Now, tsc_C_RNTI_Def2);

    //Activate RRC security at SS Cell 2 (SRB 1, 2 and 1 AM DRB)

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest (eutra_Cell2, v_Auth_Params, 0);

    f_EUTRA_Security_Set (v_Auth_Params);

    //Stop periodic sending of TA command on source cell

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PUCCH_Synch_None));

    //Configure UL grant configuration ("OnSR", periodic TA is NOT started) in target cell (cell 2)

    f_EUTRA_SS_CommonCellConfig(eutra_Cell2, cas_ULGrantAllocation_Def_REQ(eutra_Cell2, cs_TimingInfo_Now, cs_PUCCH_Synch_None));

    //@siclog "Step 16" siclog@

    // The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform intra frequency handover to Cell 2.

    f_EUTRA_RRC_ConnReconfigHO_IntraLTE_Mobility (eutra_Cell1,

                                                  eutra_Cell2,

                                                  f_Generate_cs_508_MobilityControlInfo_HO(eutra_Cell2));

    //@siclog "Step 17" siclog@

    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message on Cell 2 with logMeasAvailable set true

    //to confirm the successful completion of the intra frequency handover?

    SRB.receive(car_SRB1_RrcPdu_IND (eutra_Cell2, cr_RRCConnectionReconfigurationComplete_r10(tsc_RRC_TI_Def, -, true_)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.2.11 Step 17");

    //Start periodic sending of TA command in target cell

    f_EUTRA_SS_CommonCellConfig(eutra_Cell2, cas_PUCCH_Synch_Config_REQ(eutra_Cell2, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def));

    //Inform the SS about completion of the HO (e.g. to trigger PDCP STATUS PDU)

    f_EUTRA_SS_PdcpHandoverCtrl(eutra_Cell2, cas_PdcpHandoverComplete_REQ(eutra_Cell2));

    //Reset SRBs and DRBs of source cell (cell 1)

    f_EUTRA_SS_SRBs_DRBs_Reset (eutra_Cell1);

    //@siclog "Step 18" siclog@

    //The SS sends a UEInformationRequest message with logMeasReportReq present.

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell2,

                                 cs_TimingInfo_Now,

                                 cs_508_UEInformationRequest(tsc_RRC_TI_Def)));

    //@siclog "Step 19" siclog@

    //Check: Does the UE send a UEInformationReponse message with logMeasReport including logged measurements did on Cell 1 and Cell 12?

    SRB.receive(car_SRB2_RrcPdu_IND(eutra_Cell2,

                                    cr_UEInformationResponse_LogMeasInfo_2Entries(tsc_RRC_TI_Def, v_AbsoluteTimeInfo,

                                                                                  v_PLMN_Identity_Cell1,'0000EF'O, '001A'O, '05'O,

                                                                                  *, cr_Cgi(v_PLMN_Identity_Cell12, v_CellIdentity_Cell12),

                                                                                  *, cr_Cgi(v_PLMN_Identity_Cell1, v_CellIdentity_Cell1),

                                                                                  cr_MeasResultNeighCells_r10_EUTRA_1Entry(v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq,

                                                                                                                           v_PhysCellId_Cell1,

                                                                                                                           *),

                                                                                  cr_MeasResultNeighCells_r10_EUTRA_1Entry(v_Frequency_IE_f2.UL_DL_Earfcn.dl_CarrierFreq,

                                                                                                                           v_PhysCellId_Cell12,

                                                                                                                           *)                                                                  )));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.2.11 Step 19");

    f_EUTRA_TestBody_Set (false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell2, E2_CONNECTED);

  } // end of f_TC_8_6_2_11_EUTRA


After:
New function:
	  import from EUTRA_CellInfoFrequency all;

  import from EUTRA_Parameters all;

  import from EUTRA_BandDependentParam all;

  import from EUTRA_SysInfo_Templates all;

// SIB 5 configuration

  function fl_InitSIB5Info_eMDT(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC return template (value) SystemInformationBlockType5

  {

    var Ul_Bandwidth_Type v_UL_Bandwidth := f_EUTRA_CellInfo_GetUL_ChBandwidth (p_CellId);

    var Dl_Bandwidth_Type v_DL_Bandwidth := f_EUTRA_CellInfo_GetDL_ChBandwidth (p_CellId);

    var Frequency_fList_Type v_Frequency_PrimaryBand := f_EUTRA_InitFrequency_f1Tof4 (px_ePrimaryFrequencyBand, v_DL_Bandwidth, v_UL_Bandwidth);

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam ( v_DL_Bandwidth, v_UL_Bandwidth );

    var CarrierFreqEUTRA v_EUTRA_CarrierFreq1;

    var CarrierFreqEUTRA v_EUTRA_CarrierFreq2;

    var template (value) InterFreqCarrierFreqList v_InterFreqCarrierList;

        if (v_Frequency_PrimaryBand.f3.dl_CarrierFreq != tsc_DummyFrequency) {    // f3 is defined for the Band px_ePrimaryFrequencyBand

          v_InterFreqCarrierList := {

            cs_InterFreqCarrierFreq(v_Frequency_PrimaryBand.f1, v_ChannelBandwidthDependency.AllowedMeasBandwidth),

            cs_InterFreqCarrierFreq(v_Frequency_PrimaryBand.f3, v_ChannelBandwidthDependency.AllowedMeasBandwidth)

          };

        } else {  // f3 not defined for the band

          v_InterFreqCarrierList := {

            cs_InterFreqCarrierFreq(v_Frequency_PrimaryBand.f1, v_ChannelBandwidthDependency.AllowedMeasBandwidth)

          };

        }

    return cs_508_SystemInformationBlockType5_Def(v_InterFreqCarrierList);

}


	  function f_TC_8_6_2_11_EUTRA() runs on EUTRA_PTC

  { //Logged MDT / Logging and reporting / Reporting at intra LTE handover / PLMN list

    var template (value) CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) CellPowerList_Type v_CellPowerList_AtT2;

    var Frequency_IE_Type v_Frequency_IE_f1;

    var Frequency_IE_Type v_Frequency_IE_f2;

    var EUTRA_CellInfo_Type v_CellInfo_Cell1;

    var EUTRA_CellInfo_Type v_CellInfo_Cell12;

    var CellIdentity v_CellIdentity_Cell1;

    var CellIdentity v_CellIdentity_Cell12;

    var PLMN_Identity v_PLMN_Identity_Cell1;

    var PLMN_Identity v_PLMN_Identity_Cell12;

    var PhysCellId v_PhysCellId_Cell1, v_PhysCellId_Cell12;

    var float v_TimerValue1;

    var float v_TimerValue2;

    var B48_Type v_AbsoluteTimeInfo;

    // var template (value) MobilityControlInfo v_MobilityControlInfo;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    var SRB_COMMON_IND v_ReceivedAsp;    //change 4
    var LogMeasInfo_r10 v_LogMeasInfo_r10; //change 4
    var integer v_FieldCnt; //change 4
    var integer i;  //change 4
    v_CellPowerList_AtT0 := {

      cs_CellPower (eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell12, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_AtT1 := {

      cs_CellPower (eutra_Cell1, -79)

    };

    v_CellPowerList_AtT2 := {

      cs_CellPower (eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),

      cs_CellPower (eutra_Cell2, -79)

    };

    // Init variables

    f_EUTRA_Init(c3);

    f_EUTRA_ChangeSIB5_Combination3 (eutra_Cell12, fl_InitSIB5Info_eMDT (eutra_Cell12));//change 1
    v_Frequency_IE_f1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);

    v_Frequency_IE_f2 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell12);

    v_CellInfo_Cell1 := f_EUTRA_CellInfo_Get(eutra_Cell1);

    v_PhysCellId_Cell1 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);

    v_CellIdentity_Cell1 := v_CellInfo_Cell1.CellIds.CellIdentity;

    v_PLMN_Identity_Cell1 := v_CellInfo_Cell1.NAS_Parameters.Guti_Parameters.PLMN_Identity;

    v_CellInfo_Cell12 := f_EUTRA_CellInfo_Get(eutra_Cell12);

    v_CellIdentity_Cell12 := v_CellInfo_Cell12.CellIds.CellIdentity;

    v_PLMN_Identity_Cell12 := v_CellInfo_Cell12.NAS_Parameters.Guti_Parameters.PLMN_Identity;

    v_PhysCellId_Cell12 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell12);

    v_TimerValue1 := f_EUTRA_SetTimerToleranceMax(eutra_Cell12, nonProtocolTimer, 30.0);

    v_TimerValue2 := f_EUTRA_SetTimerToleranceMax(eutra_Cell1, nonProtocolTimer, 30.0);

    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -79);

    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell2, -79);

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    f_EUTRA_CellConfig_Def(eutra_Cell12);

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);

    //The UE is registered on PLMN1 (Cell 1) using the procedure described in TS 36.508[18]

    //clause 4.5.2.3 except that the ATTACH ACCEPT message indicates PLMN2 in

    //the Equivalent PLMN list as described in Table  8.6.2.11.3.3-5

    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell12);

    //Bring UE to inital state with Equivalent PLMN2

    f_EUTRA_IdleUpdated_WithEplmn (eutra_Cell12,

                                   PREAMBLE,

                                   cs_PLMN_List_1PLMN(f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell1))));

    f_EUTRA_RbEst_Def(eutra_Cell12);

    f_EUTRA_TestBody_Set (true);

    //@siclog "Step 1" siclog@

    //The SS transmits a LoggedMeasurementConfiguration message on Cell 12.

    v_AbsoluteTimeInfo := f_MDT_GetAbsolute_TimeStamp();

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell12,

                                 cs_TimingInfo_Now,

                                 cs_LoggedMeasurementConfiguration_with_v1080_IEs(v_PLMN_Identity_Cell12, '0000EF'O, '001A'O, '05'O, v_AbsoluteTimeInfo, -, -, -,

                                                                                  cs_LoggedMeasurementConfiguration_v1080_IEs({v_PLMN_Identity_Cell1, v_PLMN_Identity_Cell12}))));

    //@siclog "Step 2" siclog@

    //The SS transmits an RRCConnectionRelease message to release the RRC connection.

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell12);

    //@siclog "Step 3" siclog@

    //Wait 30 secods for UE performing the logging at regular time intervals on Cell 12.

    f_Delay(v_TimerValue1);

    //@siclog "Step 4" siclog@

    //The SS changes Cell 1 and Cell 12 levels according to the row "T1" in table 8.6.2.11.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 5" siclog@

    //The generic test procedure in TS 36.508 subclause 6.4.2.7 is performed on Cell 1.

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_EUTRA_TrackingAreaUpdate(eutra_Cell1);

    //@siclog "Step 6" siclog@

    //Wait 30 secods for UE performing the logging at regular time intervals on Cell 1

    f_Delay(v_TimerValue2);

    //@siclog "Step 7 - 14" siclog@

    // Steps 2 to 9 of the generic radio bearer establishment procedure in TS 36.508 subclause 4.5.3.3 are executed to successfully complete the service request procedure on Cell 1.

    f_EUTRA_RbEst_Def(eutra_Cell1);

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);//Change 2
    //Get security parameters

    v_Auth_Params := f_EUTRA_Security_Get ();

    //Inform the SS about the HO and about the source cell id

    f_EUTRA_SS_PdcpHandoverCtrl(eutra_Cell2, cas_PdcpHandoverInit_REQ(eutra_Cell1, eutra_Cell2)); //change 3
    //Configure C-RNTI based contention resolution in cell 2

    f_EUTRA_SS_ConfigRachProcedure_HO (eutra_Cell2, cs_TimingInfo_Now, tsc_C_RNTI_Def2);

    //Activate RRC security at SS Cell 2 (SRB 1, 2 and 1 AM DRB)

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest (eutra_Cell2, v_Auth_Params, 0);

    f_EUTRA_Security_Set (v_Auth_Params);

    //Stop periodic sending of TA command on source cell

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PUCCH_Synch_None));

    //Configure UL grant configuration ("OnSR", periodic TA is NOT started) in target cell (cell 2)

    f_EUTRA_SS_CommonCellConfig(eutra_Cell2, cas_ULGrantAllocation_Def_REQ(eutra_Cell2, cs_TimingInfo_Now, cs_PUCCH_Synch_None));

    //@siclog "Step 16" siclog@

    // The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform intra frequency handover to Cell 2.

    f_EUTRA_RRC_ConnReconfigHO_IntraLTE_Mobility (eutra_Cell1,

                                                  eutra_Cell2,

                                                  f_Generate_cs_508_MobilityControlInfo_HO(eutra_Cell2));

    //@siclog "Step 17" siclog@

    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message on Cell 2 with logMeasAvailable set true

    //to confirm the successful completion of the intra frequency handover?

    SRB.receive(car_SRB1_RrcPdu_IND (eutra_Cell2, cr_RRCConnectionReconfigurationComplete_r10(tsc_RRC_TI_Def, -, true_)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.2.11 Step 17");

    //Start periodic sending of TA command in target cell

    f_EUTRA_SS_CommonCellConfig(eutra_Cell2, cas_PUCCH_Synch_Config_REQ(eutra_Cell2, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def));

    //Inform the SS about completion of the HO (e.g. to trigger PDCP STATUS PDU)

    f_EUTRA_SS_PdcpHandoverCtrl(eutra_Cell2, cas_PdcpHandoverComplete_REQ(eutra_Cell2));

    //Reset SRBs and DRBs of source cell (cell 1)

    f_EUTRA_SS_SRBs_DRBs_Reset (eutra_Cell1);

    //@siclog "Step 18" siclog@

    //The SS sends a UEInformationRequest message with logMeasReportReq present.

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell2,

                                 cs_TimingInfo_Now,

                                 cs_508_UEInformationRequest(tsc_RRC_TI_Def)));

    //@siclog "Step 19" siclog@

    //Check: Does the UE send a UEInformationReponse message with logMeasReport including logged measurements did on Cell 1 and Cell 12?

    //change 4

    SRB.receive(car_SRB2_RrcPdu_IND(eutra_Cell2,

                                    cr_508_UEInformationResponse(tsc_RRC_TI_Def, omit, omit, cr_LogMeasReport_r10(v_AbsoluteTimeInfo, v_PLMN_Identity_Cell12,'0000EF'O, '001A'O, '05'O, *, *))))

      -> value v_ReceivedAsp; //@sic R5s130525 sic@

    v_FieldCnt := lengthof(v_ReceivedAsp.Signalling.Rrc.Dcch.message_.c1.ueInformationResponse_r9.criticalExtensions.c1.ueInformationResponse_r9.nonCriticalExtension.nonCriticalExtension.logMeasReport_r10.logMeasInfoList_r10); //@sic R5s130525 sic@

    if (v_FieldCnt < 2) {f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 20: Not enough LogMeasurementInfo entries")} //@sic R5s130525 sic@

    for (i:=0; i<v_FieldCnt; i:=i+1) {

      v_LogMeasInfo_r10 := v_ReceivedAsp.Signalling.Rrc.Dcch.message_.c1.ueInformationResponse_r9.criticalExtensions.c1.ueInformationResponse_r9.nonCriticalExtension.nonCriticalExtension.logMeasReport_r10.logMeasInfoList_r10[i];

      if (not match(v_LogMeasInfo_r10, cr_LogMeasInfo_r10(*, ?, cr_Cgi(v_PLMN_Identity_Cell12,v_CellIdentity_Cell12), (0..97), (0..34), cr_MeasResultNeighCells_r10_EUTRA_1Entry(v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq,

                                                                                                                         v_PhysCellId_Cell1,

                                                                                                                         omit))))

        {

          if (i == v_FieldCnt - 1) {f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 19: None of the LogMeasurementInfo matched the template for cell 1")}

        }

      else

        {

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "step19 Log meas info matched for cell 1"); break;

        }

    }

      for (i:=0; i<v_FieldCnt; i:=i+1) {

      v_LogMeasInfo_r10 := v_ReceivedAsp.Signalling.Rrc.Dcch.message_.c1.ueInformationResponse_r9.criticalExtensions.c1.ueInformationResponse_r9.nonCriticalExtension.nonCriticalExtension.logMeasReport_r10.logMeasInfoList_r10[i];

      if (not match(v_LogMeasInfo_r10, cr_LogMeasInfo_r10(*, ?, cr_Cgi(v_PLMN_Identity_Cell1,v_CellIdentity_Cell1), (0..97), (0..34), cr_MeasResultNeighCells_r10_EUTRA_1Entry(v_Frequency_IE_f2.UL_DL_Earfcn.dl_CarrierFreq,

                                                                                                                         v_PhysCellId_Cell12,

                                                                                                                         omit))))

        {

          if (i == v_FieldCnt - 1) {f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 19: None of the LogMeasurementInfo matched the template for cell 12")}

        }

      else

        {

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Log meas info matched for cell 12"); break;

        }

    } 

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.2.11 Step 19");

    f_EUTRA_TestBody_Set (false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell2, E2_CONNECTED);

  } // end of f_TC_8_6_2_11_EUTRA


5. Execution Log Files

1) Samsung Shannon UE
The Samsung Shannon UE passed this test case on Anite Conformance Toolset Solution in LTE FDD path on band 3. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_8_6_2_11_UE1_Log.htm   

References

	[1]
	R5s141312:    Supporting information for agreement of LTE-A Rel-11 eMDT Test Case 8.6.2.11. This archive comprises: html and text format execution log files as well as PICS/PIXIT settings.

	
	


